NAG Fortran Library Routine Document

E01BEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E01BEF computes a monotonicity-preserving piecewise cubic Hermite interpolant to a set of data points.

2 Specification

```
SUBROUTINE E01BEF(N, X, F, D, IFAIL)
INTEGER           N, IFAIL
real              X(N), F(N), D(N)
```

3 Description

This routine estimates first derivatives at the set of data points (x_r, f_r), for $r = 1, 2, \ldots, n$, which determine a piecewise cubic Hermite interpolant to the data, that preserves monotonicity over ranges where the data points are monotonic. If the data points are only piecewise monotonic, the interpolant will have an extremum at each point where monotonicity switches direction. The estimates of the derivatives are computed by a formula due to Brodlie, which is described in Fritsch and Butland (1984), with suitable changes at the boundary points.

The routine is derived from routine PCHIM in Fritsch (1982).

Values of the computed interpolant, and of its first derivative and definite integral, can subsequently be computed by calling E01BFF, E01BGF and E01BHF, as described in Section 8.

4 References

Fritsch F N and Butland J (1984) A method for constructing local monotone piecewise cubic interpolants *SIAM J. Sci. Statist. Comput.* **5** 300–304

Fritsch F N (1982) PCHIP final specifications *Report UCID-30194* Lawrence Livermore National Laboratory

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of data points.

Constraint: N ≥ 2.

2: X(N) – ***real*** array *Input*

On entry: X(r) must be set to x_r, the rth value of the independent variable (abscissa), for $r = 1, 2, \ldots, n$.

Constraint: X$(r) <$ X$(r + 1)$.

3: F(N) – ***real*** array *Input*

On entry: F(r) must be set to f_r, the rth value of the dependent variable (ordinate), for $r = 1, 2, \ldots, n$.

4: D(N) – ***real*** array *Output*

On exit: estimates of derivatives at the data points. D(r) contains the derivative at X(r).

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 2.

IFAIL $= 2$

The values of X(r), for $r = 1, 2, \ldots,$ N, are not in strictly increasing order.

7 Accuracy

The computational errors in the array D should be negligible in most practical situations.

8 Further Comments

The time taken by the routine is approximately proportional to n.

The values of the computed interpolant at the points PX(i), for $i = 1, 2, \ldots,$ M, may be obtained in the ***real*** array PF, of length at least M, by the call:

```
CALL E01BFF(N,X,F,D,M,PX,PF,IFAIL)
```

where N, X and F are the input parameters to E01BEF and D is the output parameter from E01BEF.

The values of the computed interpolant at the points PX(i), for $i = 1, 2, \ldots,$ M, together with its first derivatives, may be obtained in the ***real*** arrays PF and PD, both of length at least M, by the call:

```
CALL E01BGF(N,X,F,D,M,PX,PF,PD,IFAIL)
```

where N, X, F and D are as described above.

The value of the definite integral of the interpolant over the interval A to B can be obtained in the ***real*** variable PINT by the call:

```
CALL E01BHF(N,X,F,D,A,B,PINT,IFAIL)
```

where N, X, F and D are as described above.

9 Example

This example program reads in a set of data points, calls E01BEF to compute a piecewise monotonic interpolant, and then calls E01BFF to evaluate the interpolant at equally spaced points.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     E01BEF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX
      PARAMETER        (MMAX=50,NMAX=50)
*     .. Local Scalars ..
      real             STEP
      INTEGER          I, IFAIL, M, N, R
*     .. Local Arrays ..
      real             D(NMAX), F(NMAX), PF(MMAX), PX(MMAX), X(NMAX)
*     .. External Subroutines ..
      EXTERNAL         E01BEF, E01BFF
*     .. Intrinsic Functions ..
      INTRINSIC        MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'E01BEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.GT.0 .AND. N.LE.NMAX) THEN
         DO 20 R = 1, N
            READ (NIN,*) X(R), F(R)
   20    CONTINUE
         IFAIL = 0
*
         CALL E01BEF(N,X,F,D,IFAIL)
*
         READ (NIN,*) M
         IF (M.GT.0 .AND. M.LE.MMAX) THEN
*           Compute M equally spaced points from X(1) to X(N).
            STEP = (X(N)-X(1))/(M-1)
            DO 40 I = 1, M
               PX(I) = MIN(X(1)+(I-1)*STEP,X(N))
   40       CONTINUE
            IFAIL = 0
*
            CALL E01BFF(N,X,F,D,M,PX,PF,IFAIL)
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) '                Interpolated'
            WRITE (NOUT,*) '     Abscissa          Value'
            DO 60 I = 1, M
               WRITE (NOUT,99999) PX(I), PF(I)
   60       CONTINUE
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,F13.4,2X,F13.4)
      END
```

9.2 Program Data

```
E01BEF Example Program Data
   9                N, the number of data points
  7.99   0.00000E+0  X(R), F(R), independent and dependent variable
  8.09   0.27643E-4
  8.19   0.43750E-1
  8.70   0.16918E+0
  9.20   0.46943E+0
 10.00   0.94374E+0
 12.00   0.99864E+0
 15.00   0.99992E+0
```

```
 20.00   0.99999E+0  End of data points
  11                 M, the number of evaluation points
```

9.3 Program Results

```
 E01BEF Example Program Results

                Interpolated
     Abscissa          Value
       7.9900         0.0000
       9.1910         0.4640
      10.3920         0.9645
      11.5930         0.9965
      12.7940         0.9992
      13.9950         0.9998
      15.1960         0.9999
      16.3970         1.0000
      17.5980         1.0000
      18.7990         1.0000
      20.0000         1.0000
```